

Today: Procedures and Programs for the Analysis of the Bimbo Bakery Data

My data format

C1	thu-dels	53	0
C2	fri-dels	53	0
C3	sat-dels	53	0
C4	mon-dels	53	0
C5	tue-dels	53	0
C6	wed-dels	53	0
C7	wk-dels	53	0
C10	weeknum	53	0
C11	thu-sals	53	0
C12	fri-sals	53	0
C13	sat-sals	53	0
C14	mon-sals	53	0
C15	tue-sals	53	0
C16	wed-sals	53	0
C17	wk-sals	53	0
C21	thu-over	53	0
C22	fri-over	53	0
C23	sat-over	53	0
C24	mon-over	53	0
C25	tue-over	53	0
C26	wed-over	53	0
C27	wk-over	53	0
C30	detrend	53	0
C40	dayofyr	318	0
C41	dels-ts	318	0
C42	sals-ts	318	0
C43	over-ts	318	0

Step 1: Adjust data

Want to eliminate seasonal trend but not weekly trend. We will eventually analyze each day of the week separately, and we don't want to take away the weekly cycle.

Use the delivery data to adjust the seasonal trend – there are two reasons for this –
The sales data is censored by delivery when demand exceeds delivery. And we assume the company is trying to gear deliveries to match anticipated demand which may well have a known seasonal trend.

From now on when I refer to “actual” deliveries and sales, I mean these adjusted actuals.

Fit loess to the weekly totals (or averages). Compute mean for year ($=678/\text{wk}$). Use the factor $678/(\text{loess fit})$ to adjust both the daily deliveries and sales, by multiplying the deliveries and sales by this factor (53 such factors for the year’s data).

Step 2:

Guess the demand distribution for a particular day of the week – both model and parameters. Simulate a year of demand for this day. By comparing the simulated demand with the actual deliveries, you can calculate the sales for each day. Then we have to compare these simulated sales (SS) with the actual sales (AS). (Of course it is only the distribution that would be meaningful – the weekly pattern would not mean anything, since we have simulated demand each day.) To do this comparison, a good way is to compute the ECDF for each distribution, and overlay them in a graph. This graph shows two quantile plots. If the distributions are the same, the two quantile plots would match. (Note that if we just sort the data in each distribution and plot it against row number, we get the same picture in this case since the two sample sizes are equal – still , it is nice to know we are working with sample cdfs.) While the eye is quite good at judging the closeness of the match, you can also compute some numerical indices, like the average absolute difference or the maximum absolute difference. The program “bimbotry” does all this for each guess. The program “bimbotryrun” actually allows you to go through a series of guesses and prints out each graph of ecdfs and a table at the end.

Step 3:

Now that you have done the best you can to estimate the demand distribution, you need to go back through the year’s deliveries with your simulated demand, and compute the SS again but this time calculate the profit using the cost parameters. The program “bimboprofit” does this, and you can vary the delivery amounts by multiplying it by factors like .9, 1.1, 1.5 etc to see what modification of the delivery schedule would be most profitable, and “bimboprofitrun” does a series of these and draws a graph from which you can pick off the max profit factor. This is almost the end. One can summarize the result by suggesting that by applying this factor to the old method of determining delivery amounts changes the annual profit for this day of the week from to an increase of%!

Programs:

```
Gmacro
bimbotryrun.mac
erase c501-c504
let k9=1
```

```

do k1=90:140/10
do k2=20:50/10
%bimbotry
name c501 'GOFmean' c502 'GOFmax' c503 'meandemd' c504
'SDdemand'
let c501(k9)=k5
let c502(k9)=k6
let c503(k9)=k1
let c504(k9)=k2
let k9=k9+1
enddo
enddo
print c503 c504 c501 c502
Endmacro

```

```

Gmacro
bimbotry.mac
#retr 'bimbo.mtp'
let k7=31          #Index of delivery day desired
let k8=11          #index of sales day desired
do k10=1:100      # do 100 sales simulations
#let k1=?         #guess mean demand
#let k2=?         #guess SD demand
let k3=k2**2/k1   # find a and b for gamma
let k4=k1**2/k2**2
rand 53 c100;     #generate a year's demand with guessed dist
#norm k1 k2.
gamm k4 k3.
let k11=100+k10   # set receiving columns
rmin ck7 c100 ck11 #record simulated sales in ck11
sort ck11 ck11    #sort simulated sales
enddo
rmean c101-c200 c301 #average order statistics
                      #(for comparison with act sales)
name c301 'simsales' c302 'actsales'
sort ck8 c302     # sort actual sales
set c304         # prepare for overlay plots of empirical cdfs
1:53

```

```

end
name c304 'ECDF'
let c304=c304/53
plot c304*c301 c304*c302;
overlay;
connect.
#plot c302*c301          # like a q-q plot
let c303=abs(c302-c301)
mean c303 k5           # an index of equality of order statistics
maxi c303 k6           # another index -small is good.
endmacro

```

```

Gmacro
bimboprofitrun.mac
erase c306 c307
Do k25=1:6
let k15=1+(k25-1)/5
%bimboprofit
Let c306(k25)=k16
Let c307(k25)=k15
enddo
name c306 'profit2'
name c307 'factor2'
plot c306*c307;
connect.
endmacro

```

```

Gmacro
bimboprofit.mac
brief 0
#let k15=1          # factor to modify deliveries
let k7=31           #delivery column index
let c404=k15*ck7   #trial delivery schedule
name k15 'factor'
let k1=140          #mean demand
let k2=30           #SD demand

```

```

let k3=k2**2/k1      # find a (=k4) and b (=k3) for gamma
let k4=(k1**2)/(k2**2)
do k10=1:100        # do 100 sales simulations
rand 53 c400;       #generate a year's demand with guessed dist
gamm k4 k3.
#norm k1 k2.
name c400 'demand1'
name c401 'simsls1'
name c402 'overage1'
name c403 'lost1'
name c404 'deliver1' #proposed new delivery amount
rmini c404 c400 c401 #put simulated sales in c401
let c402 = c404-c401 #overage in c402
let c403 = (c400-c401 +abs(c400-c401))/2 #lost sales
name k11 'nsales' k12 'nover' k13 'nlost'
sum c401 k11        # total sales for year (this day)
sum c402 k12        # total overage for year (this day)
sum c403 k13        # total unmet demand for year (this day)
let k14= .50*k11 - 2*k13-.25*k12 # year's profit
name k14 'profit'
let c405(k10)=k14   # store profit from this simulated year of this
day
enddo
name c405 'profit1'
brief 2             #restore output
mean c405 k16      # mean profit
#dotp c405         #profit distribution
endmacro

```